Lagrange Multipliers

Rob Krueger, Concordia University, St. Paul, rkrueger@csp.edu

Bret Taylor, Concordia University, Irvine, bret.taylor@cui.edu

```
> restart: with(plots):
```

We will find the global extrema of f(x,y) = $xy + 3$ with the domain restricted to the points on the ellipse g(x,y) = $4x^2 + y^2 = 4$ (or g(x,y) = $4x^2 + y^2 - 4 = 0$).

We first plot the f(x,y) in yellow, the ellipse in red, and the function values of f(x,y) on the ellipse domain contraint in blue.
Note: we will graph the ellipse parametrically.

```
> functionf := plot3d(x·y + 3, x=-2..2, y=-2..2, view=0..5, axes=boxed, color=yellow) :

> functiong := spacecurve([cos(t), 2· sin(t), 0], t=0..2·Pi, axes=boxed, color=red, thickness=2) :

> values := spacecurve([cos(t), 2·sin(t), 2·cos(t)·sin(t) + 3], t=0..2·Pi, color=blue, thickness=2) :

> display(functionf, functiong, values, transparency=0.5);
```



To find these extreme values, we first graph the contour plot of f(x,y) and the ellipse constraint.

```
> cplot:=contourplot(x*y+3,x=-2..2, y=-2..2, contours=34):
> myellipse:=implicitplot(4*x^2+y^2=4, x=-2..2, y=-2..2,thickness=2):
> display(cplot,myellipse);
```



>

Notice that the (near) intersection of the ellipse and the largest (darkest red) contour and smallest (lightest yellow) contour of f(x,y) occur where the tangent lines are parallel. This means the gradient vectors are parallel. Therefore to find the values, we solve the equation:

$$\nabla f(x, y) = \lambda \nabla g(x, y) \quad \text{and} \quad 4x^2 + y^2 = 4$$

In other words:

$$f_x - \lambda g_x = 0$$

$$f_y - \lambda g_y = 0$$

$$4x^2 + y^2 - 4 = 0$$

We call λ the Lagrange Multiplier.

Now we will compute the values. First define the functions f and g.

```
> f := (x, y) -> x*y+3;
  g := (x, y) -> 4*x^2 + y^2 - 4;
```

$$f := (x, y) \to x\,y + 3$$

$$g := (x, y) \to 4\,x^2 + y^2 - 4$$

Then we define L[x] and L[y] as the left-hand side of the first two equations we wish to solve above.

```
> L[x] := diff(f(x,y),x) - lambda*diff(g(x,
  y),x);
  L[y] := diff(f(x,y),y) - lambda*diff(g(x,
  y),y);
```

$$L_x := y - 8\,\lambda\,x$$

$$L_y := x - 2\,\lambda\,y$$

Finally, we solve the system of equations made up of the constraint and the relations on the partial derivatives.

```
> solve({g(x,y), L[x], L[y]}, {x, y,
  lambda});
```

$$\left\{y = \mathrm{RootOf}\left(\_Z^2 - 2\right), \lambda = \frac{1}{4}, x = \frac{1}{2}\,\mathrm{RootOf}\left(\_Z^2 - 2\right)\right\}, \left\{\lambda = -\frac{1}{4}, x = -\frac{1}{2}\,\mathrm{RootOf}\left(\_Z^2 - 2\right), y = \mathrm{RootOf}\left(\_Z^2 - 2\right)\right\}$$

This gives four solutions that we can plug in to f(x,y) to find which points produce maximums and which points produce minimums.

```
> 'f(1/sqrt(2),sqrt(2))' = f(1/sqrt(2),sqrt
  (2));
  'f(1/sqrt(2),-sqrt(2))' = f(1/sqrt(2),-
  sqrt(2));
  'f(-1/sqrt(2),sqrt(2))' = f(-1/sqrt(2),
  sqrt(2));
  'f(-1/sqrt(2),-sqrt(2))' = f(-1/sqrt(2),-
  sqrt(2));
```

$$f\left(\frac{1}{\sqrt{2}}, \sqrt{2}\right) = 4$$

$$f\left(\frac{1}{\sqrt{2}}, -\sqrt{2}\right) = 2$$
$$f\left(-\frac{1}{\sqrt{2}}, \sqrt{2}\right) = 2$$
$$f\left(-\frac{1}{\sqrt{2}}, -\sqrt{2}\right) = 4$$

Thus we have a maximum value of 4 at $\left(\frac{1}{\sqrt{2}}, \sqrt{2}\right)$ and $\left(-\frac{1}{\sqrt{2}}, -\sqrt{2}\right)$ and a minumum value of 2 at $\left(\frac{1}{\sqrt{2}}, -\sqrt{2}\right)$ and $\left(-\frac{1}{\sqrt{2}}, \sqrt{2}\right)$.

Let see plot those points on our graphs.
In 2D: the maximums are black and the minimums are blue.

```
> mypoints:=pointplot([[1/sqrt(2),sqrt(2)],
  [-1/sqrt(2),sqrt(2)],[1/sqrt(2),-sqrt(2)],[-1/sqrt(2),-sqrt
  (2)]],color=[black,blue,blue,black],axes=boxed,
  symbol=solidbox,symbolsize=20):
> display(myellipse,mypoints,cplot);
```



In 3D: the maximums are black and the minimums are blue.

```
> mypoints3d:=pointplot3d([[1/sqrt(2),sqrt
  (2),4],[-1/sqrt(2),sqrt(2),2],[1/sqrt(2),
  -sqrt(2),2],[-1/sqrt(2),-sqrt(2),4]],
  color=[black,blue,blue,black],axes=boxed,
  symbol=solidbox,symbolsize=20):
```

```
> display(functionf,functiong,values,
  mypoints3d,transparency=0.3);
```



Indeed the maxima and minima values appear to be correct!

Checking our work with the built in package:

```
> with(Student[Calculus1]):
  with(Student[MultivariateCalculus]):
> LagrangeMultipliers(f(x,y), [g(x,y)], [x,
  y], output = detailed);
  LagrangeMultipliers(f(x,y), [g(x,y)], [x,
  y]);
```

$$\left[x=\frac{1}{2}\operatorname{RootOf}(\_Z^2-2), y=\operatorname{RootOf}(\_Z^2-2), \lambda_1=\frac{1}{4}, x\,y+3=\frac{1}{2}\operatorname{RootOf}(\_Z^2-2)^2+3\right], \left[x=-\frac{1}{2}\operatorname{RootOf}(\_Z^2-2), y=\operatorname{RootOf}(\_Z^2-2), \lambda_1=-\frac{1}{4}, x\,y+3=\right.$$

$$-\frac{1}{2}\,\mathrm{RootOf}\left(\_Z^2-2\right)^2+3\Big]$$

$$\left[\frac{1}{2}\,\mathrm{RootOf}\left(\_Z^2-2\right),\mathrm{RootOf}\left(\_Z^2-2\right)\right],\left[-\frac{1}{2}\,\mathrm{RootOf}\left(\_Z^2-2\right),\mathrm{RootOf}\left(\_Z^2-2\right)\right] \quad \textbf{(1)}$$

>

>

>